Spreadsheets, from the first click

LHUM 400 hands-on guide. You bring a question. The grid holds the numbers and shows you the answer.

> The working idea for this whole guide: a spreadsheet puts light on a question, so the action you take next is a smarter one. Data first, decision second.

This is the print companion to the interactive walkthrough page, where the grids really work. On paper, build each example in a blank Google Sheet (sheets.google.com) as you read.

Level 0: What is this thing?

A spreadsheet is a grid of boxes. Each box can hold a word or a number. And any box can do math using the other boxes.

People kept grids like this on paper for centuries: one row per thing, one column per fact about it. Rent here, groceries there, total at the bottom. The digital version keeps the same shape and adds one superpower: when a number changes, every total that depends on it updates itself. Instantly. Without being asked.

If you have never opened a spreadsheet in your life, you are in exactly the right place. This guide assumes nothing.

Level 0: Why bother?

Your head can hold maybe four numbers before they start trading places. A grid holds all of them, in daylight, at once.

But the math is not really the point. The point is the question underneath it. Can I afford this move? Which gig is actually worth my Saturday? Where does my money go every month? Questions like these feel foggy when they live in your head. Put the numbers in a grid and the fog lifts: the question becomes something you can look at, poke, and answer.

Step 1: Every box has an address

Columns are letters, rows are numbers. A box's address is column plus row, like B2. Think of a seat at a show: section B, row 2. The address is how you find one box among many, and later it is how formulas find each other.

Step 2: Type your first number

Click a box, type 42, press Enter. Two things happen: the number gets saved, and the selection drops down one row, ready for the next entry. That little hop is why data entry goes fast.

Type a word instead and the grid treats it differently: words sit on the left of the box, numbers on the right. The grid already knows which is which. Made a mess? Click the box and press Delete, or just retype. Nothing here is precious.

Step 3: A box does one of three jobs

Every spreadsheet ever built, at any size, is three jobs arranged on a grid. In the figures below, each job keeps its own color:

	A	B
1	*Label*	*names things, no math*
2	**Input**	*a number you can change*
3	**Formula**	*a calculation, starts with =*

A worked example: Friday and Saturday earnings, and a box that adds them.

	A	B
1	*Friday*	**100**
2	*Saturday*	**150**
3	*Weekend*	**250**

`B3 holds =B1+B2 and shows 250`

Change Friday's 100 and the Weekend box updates itself. You never touch B3 again.

Step 4: Formulas start with =

The equals sign tells a box to calculate instead of just holding what you typed. In slow motion, pressing Enter on =B1+B2 does three things:

1. The box reads the recipe: go get B1, go get B2, add them.

2. It fetches those values and does the math.

3. It shows the result, and it remembers the recipe.

That last part matters. The box does not store the answer. It stores the recipe, and re-cooks it any time an ingredient changes. The four basics: =B1+B2 adds, =B1-B2 subtracts, =B1*3 multiplies, =B1/2 divides.

> Point, don't type. =8+5 also gives 13, but it is a dead end: change your numbers and it sits there, wrong. =B1+B2 points at the boxes, so it stays true no matter what you put in them. Pointing is the whole trick.

Step 5: Add a whole column with a range

A range is a run of boxes: B1:B3 means B1 through B3. Feed it to SUM.

	A	B	C
1	*Mon*	**10**	**60**
2	*Tue*	**20**	
3	*Wed*	**30**	

`C1 holds =SUM(B1:B3) and shows 60`

Step 6: Ask a column some questions

SUM has cousins, and each answers a different question about the same pile of numbers. With practice minutes for a week in B1:B5:

```
=AVERAGE(B1:B5) what's typical? =MIN(B1:B5) what's the worst?
=MAX(B1:B5) what's the best? =COUNT(B1:B5) how many entries?
```

Try 20, 45, 0, 30, 25. The average is 24. Change the 0 to 60 and the average jumps. One unusual day can drag a whole average around. That is worth knowing before you trust any average, including the ones in headlines.

Step 7: Teach a box to decide with IF

IF asks a yes-or-no question and shows one answer or the other: =IF(test, answer if yes, answer if no). The test uses comparison signs: > more than, < less than, >= at least, <= at most. Words in a formula go inside straight double quotes.

	A	B	C
1	*Left over*	**250**	**OK**

```
C1 holds =IF(B1>=0, "OK", "OVER") and shows OK
```

Change B1 to -40 and the verdict flips to OVER. A box that watches a number and raises a flag when it crosses a line: that is the seed of every dashboard you have ever seen.

The ladder

Everything below is a real question someone in this course could be asking this semester. Each one starts foggy and ends as a number you can act on. Climb until one of them bites.

Scenario 1: The pizza question

What can the band afford?

Four players, one hour, 15 dollars per player per hour. Two formulas take you from the gig to the pizza order.

	A	B
1	*Players*	**4**
2	*Hours*	**1**
3	*$ per player/hr*	**15**
4	*Total earned*	**60**
5	*Pizzas at $18*	**3.33**

```
B4 holds =B1*B2*B3 and shows 60
B5 holds =B4/18 and shows 3.33
```

Change Hours to 3 and both answers jump, and nobody redid any math. That is the machine working for you.

Scenario 2: A one-month budget

Where does the month actually go?

Money in, money out, what's left.

	A	B
1	*Income*	**1200**
2	*Rent*	**700**
3	*Food*	**250**
4	*Spent*	**950**
5	*Left over*	**250**

```
B4 holds =SUM(B2:B3) and shows 950
B5 holds =B1−B4 and shows 250
```

Left over reads from Spent, which reads from the expenses. Change Rent and the change flows all the way down. Chains like this are how big models are built: each box asks the one before it.

Scenario 3: Forecast a savings goal

When can I actually have the gear?

You want 600 dollars for gear, saving 40 a week.

	A	B
1	*Save per week*	**40**
2	*Weeks*	**12**
3	*Saved so far*	**480**
4	*Goal*	**600**
5	*Still need*	**120**

```
B3 holds =B1*B2 and shows 480
B5 holds =B4−B3 and shows 120
```

Change Weeks until Still need hits 0. You just ran a forecast: same grid, pointed at the future instead of the past.

Scenario 4: The subscription audit

What do the little monthly charges really cost?

Small monthly numbers hide their real size. Multiply by 12 and they come out of hiding.

	A	B
1	*Streaming*	**12**
2	*Cloud storage*	**3**
3	*Sample packs*	**10**
4	*Per month*	**25**
5	*Per year*	**300**

```
B4 holds =SUM(B1:B3) and shows 25
B5 holds =B4*12 and shows 300
```

300 dollars is a decent microphone. Cross out one row and watch the year shrink. If you just felt a little tug about which one to cut, notice what happened: the data asked you a question you were not asking yourself.

Scenario 5: Compare two options side by side

Which gig actually pays better?

Gig A pays 120 dollars for a 3-hour night, door to door. Gig B pays 200 but eats 6 hours with the drive. One column per option, and the same question asked of both.

	A	B	C
1	*Pay*	**120**	**200**
2	*Hours (door to door)*	**3**	**6**
3	*Per hour*	**40**	**33.33**

```
B3 holds =B1/B2 and shows 40
C3 holds =C1/C2 and shows 33.33
```

The bigger check lost. The flat fee was hiding the drive, and the per-hour number put light on it. This is the pattern for comparing anything: apartments, laptops, tour routings. One column per option, one row per fact, one formula asked of every column.

Scenario 6: Break-even, with a verdict

How many tickets until the show pays for itself?

You are thinking about renting a room for a show. The room holds 40 people. Three formulas take you from costs to a straight answer.

	A	B
1	*Room*	**150**
2	*Sound tech*	**90**
3	*Ticket price*	**8**
4	*Total cost*	**240**
5	*Tickets to break even*	**30**
6	*Verdict*	**GO**

```
B4 holds =B1+B2 and shows 240
B5 holds =B4/B3 and shows 30
B6 holds =IF(B5<=40, "GO", "RISKY") and shows GO
```

Raise the room to 250 and the verdict flips. The sheet is not deciding for you. It is showing you exactly where the line sits, so you can argue with the inputs instead of with your gut. Want the show to work? Now you know your levers: a cheaper room, a higher ticket, or a bigger venue.

Scenario 7: Cost per year of owning

Is the cheap one actually cheaper?

A 60-dollar audio interface that dies in 2 years, or a 180-dollar one that runs for 8? The sticker answers what leaves your pocket today. Cost per year answers what owning it really costs. Different question, sometimes a different winner.

	A	B	C
1	*Price*	**60**	**180**
2	*Years it lasts*	**2**	**8**
3	*Cost per year*	**30**	**22.50**

```
B3 holds =B1/B2 and shows 30
C3 holds =C1/C2 and shows 22.50
```

The trick generalizes: cost per use, cost per gig, cost per student, cost per mile. Whenever two options refuse to compare, divide both by the thing you actually care about and they will.

Sandbox: Ask your own question

Pick a question you actually have and rough it out in a blank sheet. Some starters: what would moving to a cheaper place really save in a year? What does my coffee habit cost per semester? How many lessons until the new amp is paid off? Which of two summer plans leaves me with more in September?

Shape to aim for: labels down the left, inputs you can change, formulas that point at them, and if you want a straight answer, an IF verdict at the bottom.

Example sheets: real spreadsheets to open and dissect

Reading other people's spreadsheets is how you get good, the same way listening is how you learn to play. Everything below is free, lives in (or copies into) Google Sheets, and asks for no signup. Open one and go hunting for the three jobs: labels, inputs, formulas.

Simple. Google's own Monthly Budget template: open sheets.google.com, click Template gallery, choose Monthly budget, then click the big total on the summary tab and read its formula. Also the Spreadsheet Class library (spreadsheetclass.com/templates): two dozen small free templates; copy one, break it on purpose, fix it.

Intermediate. The Measure of a Plan budget tracking tool (themeasureofaplan.com/budget-tracking-tool): a genuinely free tracker with a self-building dashboard. Trace one chart back to the cells it reads.

Advanced. Aspire Budgeting (aspirebudget.com): a full zero-based budgeting system in one Google Sheet; figure out why it is split across tabs. And the Vertex42 loan amortization schedule (vertex42.com/ExcelTemplates/loan-amortization-schedule.html): find the PMT function, change one input, and explain why the payoff date moved.

Not sure which tier is yours? The level quiz on the course site points you at a starting line.

Next: powers the real thing adds

Once your model works, Google Sheets (or Excel, or Numbers) gives you more. Charts: select your numbers, then Insert > Chart, and the picture updates with the data. Sorting and filtering, for when your rows number in the hundreds. Sharing, so a bandmate or roommate can argue with your inputs instead of your conclusions.

> Same three jobs, same =, same colon for ranges. Sheets, Excel, and Numbers are one grid in different clothes.

A box can label, hold, or calculate. That's the whole secret. The rest is asking better questions.